
The Clipper 5 feature that receives the most blank stares from xBase programmers climbing 

aboard theClipper bandwagon has got to be code blocks. The problem isn't with understanding the 

definition. 

 

Staunch xBase developers can learn new tricks and after an introduction to code blocks, the 

consensus is usually "Hey, that's pretty cool!" The problem is one of application. Even after 

programmers learn the ins and outs of code blocks, it's still difficult to apply them in everyday 

coding chores. Once mastered, some go overboard and stick code blocks everywhere; forcing the 

issue. The secret is for programmers to get to the point where they can introduce a code block and 

have it help them represent a programming problem and allow the program to run efficiently. But 

that's enough philosophy; let's get to the task at hand. 

 

What is a code block? 

 

A code block is simply a way to store one or more precompiled Clipper 5 expressions in memory 

variables. The code block is a distinct Clipper 5 data type (as we saw in the June 1992 issue, 

VALTYPE()  returns a "B" for a variable holding a code block). Once these expressions are stored, a 

code block must be evaluated. There are several ways to evaluate a code block. We'll cover one 

method this month and save others for future explorations. 

 

Think of a code block as a function; it has code, you can pass it parameters, it returns a value, and 

it has a name, i.e. the name of the memory variable you assign it to. In order to "call" this 

facsimile of a function, you "evaluate it," executing the code within. 

 

So why not just use functions and forget about code blocks? The answer lies in how code blocks 

are embedded in various Clipper 5 constructs. This will become obvious as we explore code blocks 

here and in future articles. 

 

The bottom line is, you've got to be good at applying code blocks because Clipper 5 relies on them 

everywhere. Nowhere is this more true than in STD.CH, the preprocessor file containing user-

defined command (UDC) definitions that define Clipper's syntax. Upon close examination, you'll 

see that many UDCs map various statements and clauses into code blocks that are passed as 

parameters to the low-level functions that make up Clipper 5. Moreover, to do Clipper 5 

optimized programming, you'll need to use code blocks for processing arrays, database contents, 

browse screens, and much more. Remember this: code blocks are an integral part of Clipper 5 and 

to ignore them is to ignore much of the power of the language. 

 

Definition 

 

The structure of a formal code block definition is: 

 
<MemVarName> := { | < ParmList > | <ExpressionList>   } 
 

where MemVarName is any valid variable name which shall possess the data type "code block." 

ParmList is an optional list of parameters, separated by commas, to be passed to the expressions 

contained in the code block in much the same manner as those passed to user-defined functions 

(UDFs). If there are no parameters to be passed, the {} characters must still be included. 

ExpressionList is a series of one or more expressions, separated by commas, that are compiled and 

stored in the block. These expressions may be of any kind allowed in Clipper, e.g.,  

 
nPageCnt+1, or  
nRecCnt++, or  
cCustName := LEFT(customer->last, 10)   

 

which shows that the inline assignment operator := is perfectly fine in a code block, or even a 

simple UDF call as in: CalcAvg(). Also, the IIF()  function is particularly well suited for code blocks 

since it provides the ability to represent logic in the code. (Remember statements like 



IF/ELSE/ENDIF  aren't allowed in code blocks.) The value returned by the code block once 

evaluated becomes the value of the last expression in the list. 

 

As an example, consider the following code block: 

 

bSample := { | x, y | x + y } 

 

In this example, a code block is defined with two parameters and assigned to the variable 

bSample. The operation of the code block applies the + operator to the passed parameters. The 

block finds success with any two parameters on which + is defined, e.g.  

 
"A one" + "and a two",  
1 + 2,  
CTOD('01/01/80') + 7.  

 

Evaluation 

 

Unlike regular expressions, which are complied and executed as they're encountered in the 

program, code blocks must be explicitly evaluated. Clipper 5 has the EVAL()  built-in function for 

this purpose. The calling sequence for EVAL()  is: 

 
EVAL( bBlock, <BlockParmList> ) --> vLastBlockValue  
 

where bBlock is the name of the memory variable containing the code block and BlockParmList is 

an optional list of parameters, separated by commas, that are passed to the expressions in the 

block. The return value, vLastBlockValue, is the value of the last expression of the block. 

 

Before we look at evaluating code blocks, let's first consider the following program that uses a 

regular expression sequence: 

 
#define ANNIV 5  
 
FUNCTION main 
 
LOCAL nNewSal, nYears, dHireDate, nCurSal,; 
      nNewSal := ( dHireDate := CTOD('01/01/84'), ;   
      nCurSal := 22500, ; 
      nYears  := Seniority(dHireDate), ;  
                 IIF( nYears > ANNIV, nCurSal*1.1, ;  
                 nCurSal*1.05))  
 
* 8 and 24750.0 shall be displayed  
 
? 'Years in service: ',nYears, 'New Salary: ', nNew Sal 
 
RETURN NIL 
 
FUNCTION Seniority (dBegin)  
RETURN (INT((DATE()-dBegin)/365))  

 

Here, the assignment statement involving nNewSal shows how a series of expressions can be 

grouped using a surrounding set of parentheses. The value returned from this extended set of 

expressions is simply the value of the LAST expression, in this case the IIF().  In reality, we'd 

code this process without embedding the expressions and the IIF()  call would be an IF   

statement. Nevertheless, to prove a point, we see that the embedded expressions are compiled 

and yield the proper results when compiled. Note that the expressions are executed immediately 

when the nNewSal assignment statement is encountered. 

 

Now that we're tuned into expression groups, let's look at the code block equivalent. The built-in 

function EVAL()  evaluates the expressions found in a code block and returns a value. As with any 

built-in function, EVAL()  may be invoked on a line by itself or embedded in an expression. Here's 

the code block equivalent to the previous example: 

 

  



#define ANNIV 5  
 
FUNCTION main 
   LOCAL nNewSal, nYears, dHireDate, nCurSal, bNewS al 
 
   bNewSal   := {||}   
   dHireDate := CTOD ('01/01/84'),   
   nCurSal   := 22500  
   nYears    := Seniority( dHireDate ), ;  
                IIF (nYears > ANNIV, nCurSal*1.1, n CurSal*1.05)   
 
   nNewSal := EVAL (bNewSal)  
   ? 'Years in service: ',nYears, 'New Salary: ',; nNewSal 
 
RETURN NIL 

 

The results are identical for both examples, but one uses a code block while the other doesn't. 

 

Programming applications 

 

This is the section of the article that will seem somewhat light in terms of real-life examples. 

There's a good reason for this. Most of the places where code blocks may be applied haven't been 

covered yet by this series. You'll have to stayed tuned for more insight into where you might use a 

code block. 

 

Suffice it to say we'll see more of code blocks when we cover arrays, the browse system, database 

manipulation, and other topics. For the time being, I'll present a series of simple code block 

examples given the tools we now have. 

 

* Evaluating a simple code block  
LOCAL bBlock := {|| nValue } // "nValue" not define d  
LOCAL nValue := 123  
LOCAL nResult  
nResult := EVAL (bBlock)  
? nResult // 123 
 

Note that the local variable nValue isn't yet known when the code block that depends on it is 

defined. This is fine, since by the time the code block is evaluated, a value for nValue will be 

assigned. 

 

* More on evaluation  

 
LOCAL bBlock := { || QOUT (cChar1), QQOUT (cChar2),  123 }  
LOCAL cChar1 := "Hello ",;  
      cChar2 := "world"  
 
LOCAL nResult  
 
nResult := EVAL (bBlock) // "Hello world" is displa yed by code block  
? nResult // 123 since last expression. 
 
In this example, the code block displays a message to the screen. Since the ? command can't 

appear inside a code block (it's not an expression), the QOUT() and QQOUT()  functions are used 

instead. 

 

* One final example of calling a code block  

* w/no parameters  

 
FUNCTION main  
LOCAL bBlock := { || bfunc() }  
EVAL (bBlock) 
RETURN NIL 
 
FUNCTION bfunc  
? "Hello world" 
RETURN NIL 



 

This example shows that there's more than one way to do the same thing. Much of the power of 

code blocks lie in their ability to call UDFs. 

 

Next, we see that some simple UDFs can be written as code blocks. 

 
FUNCTION main  
LOCAL bMaxBlock := { |p1,p2,p3| MAX(p1, MAX(p2,p3))  }  
? maxfunc (3, 10, 7.5) // UDF yields: 10  
* Note below, parm list follows code block name  
? EVAL (bMaxBlock, 3, 10, 7.5) // Code block // yie lds: 10 
RETURN NIL 
 
FUNCTION maxfunc (p1, p2, p3)  
RETURN MAX (p1, MAX (p2, p3))  

 

Code blocks, like UDFs, can accept a variable number of parameters as shown in the following 

example: 

 
LOCAL bBlock := { |x, y, z| QOUT(x, y, z) }  
LOCAL vRetVal 
* First pass full parameter list 
* and get: "Hello there world"  
EVAL (bBlock, "Hello", "there", "world") 
* Now leave off one of the passed parameters 
* and get: "Hello there NIL"  
vRetVal := EVAL(bBlock, "Hello", "there")  
? vRetVal // NIL, since QOUT() always RETURNS a NIL  

 

More on code block parameter passing 

 

As with UDFs, parameters are passed to code blocks "by value" which means they may not be 

changed by the block's expressions. Consider: 

 
LOCAL cActual := 'foo'  
LOCAL bBlock := { | cFormal | QOUT( cFormal ), ;  
                    cFormal := 'fu' }  
EVAL ( bBlock,  cActual ) // Pass "By Value"  
? cActual // "foo" unchanged  
EVAL (bBlock, @cActual // Pass "By Reference"  
? cActual // "fu" changed  

 

When bBlock is evaluated, a new variable cFormal, which is local to the block, is created. The 

QOUT() displays the passed value "foo," but then the formal parameter cFormal is changed to "fu." 

Upon return however, cActual retains its original value "foo," unaltered by the evaluation of the 

code block. 

 

In the second EVAL() , the actual parameter is passed "by reference" using the @ prefix which 

allows the value to be changed inside the code block. 

 

An understanding of scope is important when defining and evaluating code blocks. Consider: 

 
LOCAL bBlock1 := { || QOUT( nParm ) }  
LOCAL bBlock2 := { | nParm | nParm++, QOUT( nParm ) , EVAL(bBlock1) }  
EVAL (bBlock2, 1)  

 

The main program passes a 1 to bBlock2 during evaluation. This actual parameter is mapped to 

nParm which is a variable local to bBlock2. Next, bBlock1 is evaluated without any parameters, but 

inside of bBlock1 is a reference to nParm. But of course, nParm is only known inside of bBlock2 and 

is therefore out of scope in bBlock1. The program aborts because nParm is not known inside of 

bBlock1. 

 

 

 

 



GET/SET block functions 

 

Clipper 5 has three new functions designed for use with code blocks, FIELDBLOCK(), 
FIELDWBLOCK() and MEMVARBLOCK(). These are presented below in standard function declaration 

notation where the parameters and return value are defined in Hungarian notation. 

 
FIELDBLOCK() --> bFieldBlock 
 

This function accepts a character parameter which contains the name of a database field opened in 

the current work area whose value you wish to retrieve (get) or alter (set). These functions are 

called GET/SET functions. For example, 

 
FIELD zip  
LOCAL bZipAgain := { || zip }  
LOCAL cField := 'ZIP', bZipBlock 
 
* Open CUSTOMER.DBF containing ZIP  
USE customer  
bZipBlock := FIELDBLOCK( cField ) // Create Get/Set   block  
? EVAL (bZipBlock) // '90024', using Get block  
? EVAL (bZipAgain) // '90024', using code block  
EVAL (bZipBlock, '99999-9999') // Replace using Set block  
 

Prior to code blocks, you'd have to resort to a much less efficient method using macro expansions 

as in: 

 
cField := 'ZIP'  
? &cField // Display contents of CUSTOMER-ZIP  
REPLACE &cField WITH '99999-9999' // Replace conten ts  
 
 
FIELDWBLOCK() --> bFieldBlock  

 

This function works just like FIELDBLOCK() , except that an additional numeric parameter is passed 

to indicate an alternate work area in which the referenced field name exists. In contrast, 

FIELDBLOCK()  assumes that the field is in the current work area. 

 
MEMVARBLOCK() --> bMemVarBlock 
 

As with the two previous functions, MEMVARBLOCK() returns a code block. This time however, the 

block is based on the name of a memory variable passed in cMemVar. One thing to note about 

MEMVARBLOCK(): it's only able to use private or public variables; locals and statics can't be 

referenced. 

 

In the example below, many of you may remember the old xBase method of using macros to 

simulate arrays. Here, this method is replaced with code blocks: 

 

* First, the macro method  

 

LOCAL i  

PRIVATE c  

FOR i  := 1 TO 3  // Create 3 privates and assign values  

    c := STR (i, 1) // Convert to char for macro  

    cMonth&c := i 

NEXT 

 
* Now, with code blocks  
PRIVATE cMonth1, cMonth2, cMonth3  
FOR i :=1 TO 3 
      * Remember, MEMVARBLOCK() returns a code  
      * block. EVAL() will use it to assign a 
      * value to the new private 
      EVAL (MEMVARBLOCK( 'cMonth' + STR (i, 1), I )  
NEXT 



The second method may appear more involved, but it runs much more efficiently since it avoids 

macros. 

 

Clipper 5 is the first xBase dialect to institute a philosophy of trying to break the mystique of the 

macro. It has long been agreed that using macros is an evil that should be avoided whenever 

possible. Clipper 5, with the help of code blocks, makes this possible. Code blocks are faster 

because they contain pre-compiled code whereas macros must interpret the code on-the-fly. And 

code blocks may be passed as parameters to procedures and functions. 

